

FONCTIONS DE RECHERCHE DANS UNE MATRICE

RECHERCHEV()




Figure 1 : la viscosité d'un élément de Densité 0.616 est de 2.93

La fonction **RechercheV()** (groupe **Recherche et référence** de l'onglet **Formule**) recherche une valeur dans une matrice, *uniquement de gauche à droite*¹. Ses arguments obligatoires sont :

- **Valeur cherchée** : valeur recherchée dans la 1^{ère} colonne de la matrice
- **Table matrice** : ensemble des cellules de la matrice
- **No Index Col** : numéro de la colonne contenant la valeur à renvoyer (en commençant par 1 pour la 1^{ère} colonne du tableau)

Remarques importantes

- La matrice doit être triée selon l'ordre croissant de la colonne de recherche si l'option suivante est **Vrai** (déconseillé)
- L'option **Valeur proche** permet de rechercher la valeur la plus proche (**Vrai** ou omis) ou la correspondance exacte (**Faux**) ; cette dernière option est conseillée pour la plupart des utilisations

L'exemple ci-dessus renvoie en B13 la viscosité correspondant à la densité saisie en A13 :

- **Valeur cherchée** : cellule A13
- **Table matrice** : cellules de A2 à C10
- **No Index Colonne** : 2
- L'argument **Valeur_proche** est omis, puisque le tableau est classé par ordre croissant de densité

¹ Cette limitation est contournée par la combinaison **Equiv()** et **Index()**, voir pages suivantes

EQUIV()

La fonction Equiv()² permet de connaître la position d'un élément dans une matrice.

Dans l'exemple suivant, la plage B1:B11 donne l'ordre d'arrivée à une course ; la fonction Equiv() en B14, détaillée en C14, renvoie le classement du coureur saisi en B13 :

	A	B	C	D
1		Classement		
2		Pierre		
3		Jean		
4		Annie		
5		Thierry		
6		Benoit		
7		Marion		
8		Hakim		
9		Kim		
10		Angélique		
11		Alfred		
12				
13	Nom	Annie		
14	Classement		3	=EQUIV(B13;B2:B11;0)

Figure 2 : la fonction Equiv()

Important : le 3^{ème} argument de cette fonction peut être égal à -1, 0 ou 1 ; seule la valeur 0 permet de trouver la valeur exacte, que la plage soit triée ou non ; c'est donc la valeur conseillée pratiquement dans tous les cas. Le tableau suivant détaille les caractéristiques des 3 possibilités pour cet argument :

TYPE	COMPORTEMENT
1 ou omis	La fonction EQUIV recherche la valeur la plus élevée qui est inférieure ou égale à celle de l'argument <i>valeur_cherchée</i> . Les valeurs de l'argument <i>matrice_recherche</i> doivent être placées en ordre croissant, par exemple : ...-2, -1, 0, 1, 2, ..., A-Z, FAUX, VRAI.
0	La fonction EQUIV recherche la première valeur exactement équivalente à celle de l'argument <i>valeur_cherchée</i> . Les valeurs de l'argument <i>matrice_recherche</i> peuvent être placées dans un ordre quelconque.
-1	La fonction EQUIV recherche la plus petite valeur qui est supérieure ou égale à celle de l'argument <i>valeur_cherchée</i> . Les valeurs de l'argument <i>matrice_recherche</i> doivent être placées en ordre décroissant, par exemple : VRAI, FAUX, Z-A, ..., 2, 1, 0, -1, -2, ..., et ainsi de suite.

² Cette fonction est très souvent couplée à la fonction Index() pour rechercher des données dans une matrice sans les limitations de la fonction Recherche() qui ne fonctionne que de gauche à droite, voir page suivante

EQUIV() ET INDEX()

La fonction Index() trouve dans une matrice une valeur en fonction d'une ligne et d'une colonne. Elle est très souvent associée à la fonction Equiv() pour faire des recherches dans une matrice en passant outre la limitation de la fonction RechercheV(), qui ne peut renvoyer une valeur que si elle est située à *droite* de la valeur cherchée.

Cette fonction est disponible en 2 versions, la 1^{ère} étant la plus utilisée :




Figure 3 : l'option « Matrice;n° lig;n°col » de la fonction Index

Soit le tableau suivant :

	A	B	C
1	Matricule	Nom	Age
2	125	Pierre	27
3	345	Jean	45
4	76	Annie	60
5	16	Thierry	45
6	232	Benoit	35
7	55	Marion	33
8	78	Hakim	40
9	20	Kim	56
10	17	Angélique	20
11	43	Alfred	18

Figure 4 : Cette matrice fournit le matricule, le nom et l'âge du personnel ; elle n'est triée selon aucun champ

L'idée va ici être de fournir le matricule et l'âge d'un membre du personnel en connaissant son nom :

1. La fonction Equiv() permettra de connaître le numéro de la ligne correspondant au nom recherché dans la matrice A2:C11
2. Ce numéro sera utilisé comme argument de numéro de ligne dans la fonction Index() pour renvoyer le matricule (avec l'argument de colonne 1, parce que le matricule est en 1^{ère} colonne de la matrice) ou l'âge (avec l'argument de colonne 3, parce que l'âge est en 3^{ème} colonne de la matrice)

Noter que seul l'âge pourrait être renvoyé par la fonction RechercheV(), parce que l'âge est à droite du nom dans la matrice.

Dans la figure suivante :

- le nom recherché est saisi en A14 (Annie)
- sa position (numéro de ligne dans la matrice) est renvoyée en A17 (ici, la 3^{ème} ligne)
- son âge est renvoyé par Equiv() et Index() en A20 (ici, 60 ans)
- son matricule est renvoyé par Equiv() et Index() en A23 (ici, 76)

	A	B	C	D
4	76	Annie	60	
5	16	Thierry	45	
6	232	Benoit	35	
7	55	Marion	33	
8	78	Hakim	40	
9	20	Kim	56	
10	17	Angélique	20	
11	43	Alfred	18	
12				
13	Nom			
14	Annie			
15				
16	Extraction de la position du prénom recherché			
17	3	=EQUIV(A14;B2:B11;0)		
18				
19	Recherche de l'age correspondant			
20	60	=INDEX(A2:C11;A17;3)		
21				
22	Recherche du matricule correspondant			
23	76	=INDEX(A2:C11;A17;1)		

Figure 5 : combinaison de Equiv() et de Index()

Il est évidemment possible de saisir une formule complète intégrant Equiv() dans Index() sans passer par la cellule intermédiaire A17 ; les formules deviennent :

- en A20 : =index(A2:C11;equiv(A14;B2:B11;0);3)
- en A23 : =index(A2 :C11;equiv(A14;B2:B11;0);1)

Il serait même possible d'utiliser les \$ dans la formule en A20 pour pouvoir la recopier directement en A23 : =index(\$A\$2:\$C\$11;equiv(\$A\$14;\$B\$2:\$B\$11;0);3) ; noter qu'il faudra quand même modifier manuellement le dernier argument 3 et 1.

UTILISATION D'EQUIV() POUR ÉTABLIR LA CORRESPONDANCE (OU NON) ENTRE 2 LISTES

La fonction Equiv() peut également servir à rechercher la présence (ou non) d'un élément d'une colonne dans une autre colonne.

Nous recherchons par exemple la présence (ou non) des noms de la colonne A dans la colonne B dans le tableau suivant :

	A	B	C	D	E
1	Colonne A	Colonne B		A dans B	B dans A
2	Jean	Thierry		3	#N/A
3	Noé	Jean		#N/A	2
4	Pierre	José		5	#N/A
5	Lucie	Pierre		#N/A	4
6	Arlette	Henri		#N/A	#N/A

Figure 6 : Equiv() met en évidence les correspondances entre 2 colonnes

Dans la colonne A dans B, par exemple :

- le 3 sur la ligne de Jean de la colonne A indique que ce dernier se trouve ligne 3 de la colonne B
- Le #N/A³ sur la ligne de Noé de la colonne A indique que ce dernier n'est pas présent dans la colonne B

Le tableau suivant affiche les formules utilisées (noter les \$ autorisant les recopies et le +1 pour tenir compte de la ligne des en-têtes de colonne) :

	A	B	C	D	E
1	Colonne A	Colonne B		A dans B	B dans A
2	Jean	Thierry		=EQUIV(A2;\$B\$2:\$B\$6;0)+1	=EQUIV(B2;\$A\$2:\$A\$6;0)+1
3	Noé	Jean		=EQUIV(A3;\$B\$2:\$B\$6;0)+1	=EQUIV(B3;\$A\$2:\$A\$6;0)+1
4	Pierre	José		=EQUIV(A4;\$B\$2:\$B\$6;0)+1	=EQUIV(B4;\$A\$2:\$A\$6;0)+1
5	Lucie	Pierre		=EQUIV(A5;\$B\$2:\$B\$6;0)+1	=EQUIV(B5;\$A\$2:\$A\$6;0)+1
6	Arlette	Henri		=EQUIV(A6;\$B\$2:\$B\$6;0)+1	=EQUIV(B6;\$A\$2:\$A\$6;0)+1

Figure 7 : les formules de correspondance Equiv()

³ Il est possible d'éviter ce #N/A avec la fonction SiErreur(), voir page 74